
Modernizace telefonu Tesla 
      



Anotace 

Tato práce se zabývá modernizací historického telefonu Tesla AS20 pomocí mikrokontroléru 

Arduino Nano a GSM modulu SIM900L. Cílem projektu bylo zachovat původní vzhled a 

ovládání telefonu, a zároveň umožnit jeho funkčnost v současné mobilní síti. V rámci realizace 

byl použit původní rotační číselník, reproduktor a zvonky, přičemž zvonění bylo řešeno 

servomotorem s pružinkovým kladívkem. Původní uhlíkový mikrofon byl nahrazen 

elektretovým mikrofonem, který je vhodnější pro spolupráci s moderní elektronikou. 

Součástí projektu bylo i využití 3D modelování a tisku pro vytvoření držáků komponent. 

Výsledné zařízení je schopné přijímat a uskutečňovat hovory, přičemž zachovává retro estetiku 

a kombinuje ji s moderními technologiemi. Práce ukazuje možnosti propojení staré a nové 

techniky a otevírá prostor pro další rozšíření, jako je zobrazení údajů na displeji nebo bateriové 

napájení pro přenosný provoz. 

Anotation 

This final project focuses on the modernization of the historical Tesla AS20 telephone using an 

Arduino Nano microcontroller and a SIM800L GSM module. The main goal was to preserve 

the original appearance and control elements of the phone while enabling it to function on a 

modern mobile network. The project incorporates the original rotary dial, speaker, and bells, 

with the ringing mechanism recreated using a servo motor with a spring-mounted hammer. The 

original carbon microphone was replaced with an electret microphone, which is more suitable 

for use with modern electronics. 

The project also includes the use of 3D modeling and printing to create custom mounts for the 

components. The resulting device is capable of both receiving and making phone calls while 

retaining the retro aesthetic combined with modern technology. This work demonstrates the 

potential of merging vintage hardware with current solutions and offers possibilities for further 

enhancements, such as adding a display for serial output or battery power for portable operation. 

Úvod 
Při úklidu na půdě jsem objevil starý telefon Tesla AS20 – relikvii minulosti, která kdysi 

sloužila jako běžný komunikační prostředek. Dnes už však nemá praktické využití, protože 

pevné linky jsou na ústupu a technologie pokročily směrem k mobilním sítím. Místo toho, 

abych telefon nechal zapomenutý nebo ho vyhodil, napadlo mě, že bych mu mohl vdechnout 

nový život. 

Cílem mé práce je modernizace tohoto historického zařízení tak, aby bylo opět funkční – 

tentokrát však v souladu s moderními technologiemi. K tomu využiji Arduino a GSM modul, 

které umožní telefon připojit k mobilní síti a simulovat jeho původní funkčnost. Výsledkem 

bude zařízení, které zachová retro design a ovládání původního telefonu, ale zároveň bude 

schopné uskutečňovat hovory prostřednictvím mobilní sítě. 

Tato práce se zaměří nejen na technické aspekty úpravy, jako je propojení Arduina s telefonním 

číselníkem a reproduktorem, ale i na zachování historické estetiky přístroje. Kromě toho budu 

řešit výzvy spojené s integrací moderních komponent do starého systému a programováním 

Arduina pro správné ovládání hovoru. 

 

Seznam zkratek a symbolů 
Obecné technické zkratky: 

DC – Direct Current (stejnosměrný proud) 

AC – Alternating Current (střídavý proud) 

V – Volt (jednotka napětí) 

A – Ampér (jednotka proudu) 

Hz – Hertz (jednotka frekvence) 

  



Elektronické součástky: 

LED – Light Emitting Diode (světelná dioda) 

LCD – Liquid Crystal Display (displej s tekutými krystaly) 

GSM – Global System for Mobile Communications (mobilní komunikační systém) 

MCU – Microcontroller Unit (mikrokontrolér, např. Arduino) 

PCB – Printed Circuit Board (tištěný spoj) 

UART – Universal Asynchronous Receiver-Transmitter (sériová komunikace) 

I/O – Input/Output (vstup/výstup) 

Arduino UNO – Mikroprocesorová deska založená na čipu ATmega328P 

SIM800L – GSM/GPRS modul pro komunikaci přes mobilní síť 

DMTF – Dual-tone multi-frequency (systém tónové volby) 

 

Seznam komponent 
• Arduino Nano R3, ATmega328 

• Arduino GPRS-GSM Shield SIM900  

• Plastové micro servo SG90 9g, 180° 

• Elektretový mikrofon 6050 JST-GH 10cm  

• Elektrodynamický reproduktor 

• Rotační číselník 

• Spínač se třemi kontakty 

• Stejnosměrný zdroj 

  



1. Teoretická část  
Telefon Tesla AS20 byl původně navržen pro pevnou linku a využíval impulsní volbu pro 

vytáčení čísel. Vzhledem k zastaralosti tohoto systému jsem se rozhodl telefon modernizovat 

pomocí Arduina a GSM modulu, což mu umožní fungovat v mobilní síti. Hlavním cílem je 

zachovat původní design a ovládání telefonu, ale zároveň přidat podporu pro moderní 

technologie. Projekt zahrnuje úpravu číselníku, napájení a integraci elektroniky tak, aby telefon 

mohl znovu přijímat a uskutečňovat hovory. 

1.1 Arduino jako řídící jednotka 
Arduino Nano v tomto projektu slouží jako řídicí jednotka, která zpracovává impulzy z 

rotačního číselníku, ovládá GSM modul SIM900 a řídí vyzvánění telefonu. Převádí impulzní 

volbu na DTMF signály a komunikuje s GSM modulem přes sériové rozhraní UART, což 

umožňuje uskutečňovat a přijímat hovory. Díky své kompaktní velikosti se Arduino Nano 

snadno vejde do těla původního telefonu, což usnadňuje integraci moderní elektroniky při 

zachování původního designu. 

1.2 Rotační číselník 
V projektu používám rotační číselník z původního telefonu Tesla AS20, který slouží jako hlavní 

způsob zadávání telefonního čísla. Tento číselník pracuje na principu impulsní volby, při které 

generuje sérii elektrických impulsů odpovídajících zvolenému číslu. Pomocí Arduina tyto 

impulsy zpracovávám a převádím je na digitální hodnoty, které následně odesílám do GSM 

modulu ve formě DTMF tónů. Díky tomu je možné zachovat autentický způsob vytáčení i v 

modernizované verzi telefonu. 

1.3 Zvonění 
Pro zachování původního zvuku telefonu jsem použil originální zvonky z telefonu Tesla AS20, 

které byly součástí jeho vyzváněcího mechanismu. Místo původního elektromagnetického 

pohonu však nyní zvuk vytvářím pomocí servomotoru. Tímto způsobem zůstává zachovaný 

autentický charakteristický zvuk původního telefonu. 

 



2. Praktická část 
Praktická část této práce popisuje konkrétní kroky, které vedly k modernizaci starého telefonu 

Tesla AS20 pomocí mikrokontroléru Arduino Nano a GSM modulu. Cílem bylo zachovat 

původní vzhled a ovládání telefonu, a zároveň mu umožnit plnohodnotné fungování v současné 

době, kdy již pevné linky prakticky vymizely. V této části podrobně popisuji postup připojení 

jednotlivých komponent, úpravy číselníku, způsob ovládání zvonků pomocí servomotoru a 

programování Arduina. Součástí je také schéma zapojení, řešení napájení a popis problémů, 

které se v průběhu realizace objevily, spolu s jejich řešením. 

2.1 Využití komponent ze starého telefonu 
V projektu jsem využil několik původních komponent telefonu Tesla AS20, které nejenže 

přispívají k autentickému vzhledu, ale také umožnily funkční propojení se současnou 

elektronikou. Mezi tyto části patří především plastový kryt a sluchátko, které zachovávají 

celkový design a ergonomii telefonu. Dále jsem použil původní rotační číselník, který po úpravě 

umožňuje snímání impulsů pomocí Arduina, a také reproduktor, který je zapojen do audio 

výstupu GSM modulu. Součástí modernizace bylo i zachování a znovupoužití původních 

zvonků, které nyní rozeznívá servomotor s kladívkem. Zajímavým prvkem je také využití 

původního relé, které jsem zapojil jako detektor zvednutí a zavěšení sluchátka – Arduino tak 

ví, kdy má navázat nebo ukončit hovor. Díky těmto prvkům se podařilo zachovat nejen vzhled, 

ale i řadu původních funkcí telefonu v moderní podobě.  

 

Obrázek 1– původní vnitřek telefonu 

 



2.2 Zapojení číselníku 
Pro připojení rotačního číselníku k Arduinu byla využita pouze jedna dvojice vodičů – zelený 

a žlutý, která slouží pro generování pulzů odpovídajících jednotlivým číslicím. Tyto pulzy jsou 

vyhodnocovány mikrokontrolérem přes digitální vstup pin 2. Druhá dvojice vodičů – hnědý a 

bílý, která standardně signalizuje, zda je číselník právě používán (tedy zda dochází k zadávání 

čísla), nebyla v tomto zapojení využita.  

Zapojení bylo realizováno podle níže uvedeného schématu: 

Výstup z číselníku (zelený vodič) je připojen přes odpor 470 Ω na digitální pin 2 Arduina. 

Tento vstup je dále stažen pomocí pull-down rezistoru 10 kΩ k zemi, čímž je zajištěna stabilita 

signálu při nečinnosti číselníku. 

Napájení Arduina je zajištěno přes standardní 5V výstup a GND. 

Použité zapojení odpovídá přiloženému schématu zapojení.  

Obrázek 2 – zapojení číselníku                                                                            

                                                                                                                           Obrázek 3 – rotační číselník 

2.3 Zapojení GSM modulu 
Pro komunikaci mezi GSM modulem SIM900 a Arduinem bylo použito sériové rozhraní, 

přičemž komunikace neprobíhá přes hardwarové piny TX/RX (0 a 1), ale přes softwarovou 

sériovou komunikaci, která využívá digitální piny 7 a 8.  

Zapojení: 

TXD (výstup z GSM modulu) je připojen na pin 7 Arduina (softwarový RX). 

RXD (vstup do GSM modulu) je připojen na pin 8 Arduina (softwarový TX). 



Obě zařízení mají společnou zem (GND), která je propojena mezi GSM modulem a Arduinem. 

Napájení GSM modulu je zajištěno pomocí externího zdroje 5V/2A, protože GSM modul může 

mít špičkové odběry až 2 A, což Arduino samotné nedokáže dodat. GND od napájecího zdroje 

je také spojeno s GND obvodu pro zajištění společné referenční úrovně napětí. 

 

Obrázek 4 – propojení Arduina a GSM modulu 

2.4 Zvonění 
V rámci praktické realizace jsem si v programu Fusion 360 navrhl držák na původní zvonky, 

který bylo potřeba přesně přizpůsobit vnitřnímu prostoru telefonu. Držáček zajišťuje stabilní 

uchycení zvonků a zároveň umožňuje správné umístění servomotoru tak, aby kladívko naráželo 

do zvonků ve správném úhlu. Model jsem následně vytiskl na 3D tiskárně z plastu, což mi 

umožnilo rychle vyrobit funkční a přizpůsobitelný díl. Na hřídelku servomotoru jsem připevnil 

pružinku s malým kladívkem, která při otáčení naráží do zvonků a tím simuluje původní 

vyzváněcí zvuk telefonu. Tato mechanická část tak elegantně kombinuje moderní technologii s 

autentickým zvukovým výstupem. 

  

Obrázek 5 – mechanismus zvonění 



2.5 Sluchátko 
Při zapojování reproduktoru a mikrofonu jsem narazil na problém s původním čtyřvodičovým 

kabelem, který vedl mezi tělem telefonu a sluchátkem – dva z jeho vodičů byly přerušené a 

nefunkční. Rozhodl jsem se proto využít jiný čtyřvodičový kabel, který původně sloužil jako 

přívod signálu od telefonní zásuvky. Tento kabel jsem přesměroval tak, aby vedl ze sluchátka 

zpět do těla telefonu, a přes původní svorky ve sluchátku jsem k němu připojil mikrofon a 

reproduktor. Na druhé straně tohoto kabelu jsem připájel dva vodiče ze starých sluchátek – 

jeden pro reproduktor a druhý pro mikrofon, na jejichž koncích jsou 3,5mm jack konektory, 

které jsem následně jednoduše zapojil do GSM modulu. Tímto řešením jsem zachoval původní 

vzhled telefonu a zároveň umožnil jednoduché a funkční propojení se současnou elektronikou. 

 

Obrázek 6 – sluchátko s přívodním kabelem 

  



2.5.1 Reproduktor 

V rámci zachování co největší autenticity jsem se rozhodl použít původní elektrodinamický 

reproduktor z telefonu Tesla AS20. Tento reproduktor je zapojen do modernizovaného obvodu 

a slouží k přenosu zvuku při hovoru. Díky tomu zůstává zachován nejen vzhled, ale i 

charakteristický zvukový projev původního zařízení, což přispívá k celkovému retro dojmu z 

celého projektu. 

 

Obrázek 7  – původní použitý reproduktor 

2.5.2 Mikrofon 

Původní uhlíkový mikrofon, který byl součástí telefonu Tesla AS20, jsem se rozhodl nepoužít, 

protože ke svému správnému fungování vyžaduje speciální napájení a obvodové řešení, které 

by projekt zbytečně komplikovalo. Místo toho jsem zvolil elektretový mikrofon, který je pro 

práci s Arduinem mnohem vhodnější – má nízkou spotřebu, jednoduché zapojení a dobré 

vlastnosti pro přenos hlasu. Pro jeho uchycení jsem si navrhl držáček v programu Fusion 360, 

který jsem následně vytiskl na 3D tiskárně tak, aby mikrofon dobře seděl na místě původního 

a zachoval estetiku telefonu. 

Obrázek 8 – mikrofon ve vytisknutém držadle 



2.6 Programování arduina 
Tento kód umožňuje zadávání telefonního čísla pomocí rotačního číselníku, ale jen pokud je 

stisknuté speciální tlačítko. Po zadání čísla se automaticky vytočí hovor přes GSM modul. 

Druhé tlačítko slouží k přijetí nebo ukončení hovoru. Příchozí hovor je signalizován zvoněním 

pomocí servomotoru a přijímá se rovněž tímto tlačítkem. 

2.6.1 Kód pro arduino 

#include <SoftwareSerial.h> 
 
#define LED_PIN 9      // LED dioda 
#define BUTTON_PIN 6   // Tlačítko 
#define ROTARY_PIN 2   // Číselník 
 
SoftwareSerial SIM900(7, 8); 
 
bool callActive = false; 
bool incomingCall = false; 
String phoneNumber = "+420"; // Předvolba 
int digitCount = 0; 
 
int needToPrint = 0; 
int count = 0; 
int lastState = LOW; 
int trueState = LOW; 
long lastStateChangeTime = 0; 
int cleared = 0; 
 
const int dialHasFinishedRotatingAfterMs = 1000; 
const int debounceDelay = 10; 
 
unsigned long previousMillis = 0; 
const long blinkInterval = 500; 
bool ledState = LOW; 
 
void setup() { 
  SIM900.begin(19200); 
  Serial.begin(9600); 
  pinMode(LED_PIN, OUTPUT); 
  pinMode(BUTTON_PIN, INPUT_PULLUP); 
  pinMode(ROTARY_PIN, INPUT); 
 
  Serial.println("🚀 Systém spuštěn, čekám na zadání čísla..."); 
  delay(20000); // Čas pro připojení k síti 
} 
 



void loop() { 
  checkForNumberReset(); // Kontrola tlačítka pro vymazání čísla 
  readRotaryDial(); 
  checkIncomingCall(); 
 
  if (!callActive && !incomingCall && digitCount == 9) { 
    startCall(); 
  } 
 
  if (incomingCall) { 
    blinkLEDNonBlocking(); 
  } 
 
  if (callActive && digitalRead(BUTTON_PIN) == LOW) {  
    delay(200);  // Debounce 
    if (digitalRead(BUTTON_PIN) == LOW) {  
      Serial.println("🔘 Tlačítko stisknuto - ukončuji hovor..."); 
      endCall(); 
    } 
  } 
} 
 
void checkForNumberReset() { 
  if (digitalRead(BUTTON_PIN) == LOW && digitCount > 0) {  
    delay(200);  // Debounce 
    if (digitalRead(BUTTON_PIN) == LOW) {  
      Serial.println("🗑 Číslo bylo smazáno. Můžeš zadat nové."); 
      phoneNumber = "+420"; // Reset čísla 
      digitCount = 0; 
      while (digitalRead(BUTTON_PIN) == LOW); // Čekání na uvolnění tlačítka 
    } 
  } 
} 
 
void readRotaryDial() { 
  int reading = digitalRead(ROTARY_PIN); 
 
  if ((millis() - lastStateChangeTime) > dialHasFinishedRotatingAfterMs) { 
    if (needToPrint) { 
      int digit = count % 10; 
      Serial.print(digit); 
      phoneNumber += String(digit); 
      digitCount++; 
 
      Serial.println(" 📞 Zadáno: " + phoneNumber); 
 
      if (digitCount == 9) { 
        Serial.println("✅ Číslo je kompletní: " + phoneNumber); 
        startCall();  // Automatické volání po zadání celého čísla 



      } 
 
      needToPrint = 0; 
      count = 0; 
      cleared = 0; 
    } 
  } 
 
  if (reading != lastState) { 
    lastStateChangeTime = millis(); 
  } 
 
  if ((millis() - lastStateChangeTime) > debounceDelay) { 
    if (reading != trueState) { 
      trueState = reading; 
      if (trueState == HIGH) { 
        count++; 
        needToPrint = 1; 
      } 
    } 
  } 
 
  lastState = reading; 
} 
 
void checkIncomingCall() { 
  if (SIM900.available()) { 
    String response = SIM900.readString(); 
    Serial.println(response); 
 
    if (response.indexOf("RING") != -1 && !callActive) { 
      Serial.println("📞 Příchozí hovor! LED začne blikat..."); 
      incomingCall = true; 
 
      Serial.println("⏳ Čekám na stisk tlačítka pro přijetí..."); 
      waitForButtonPress(); 
      acceptCall(); 
 
      Serial.println("⏳ Čekám na druhý stisk pro ukončení..."); 
      waitForButtonPress(); 
      endCall(); 
    } 
 
    // Kontrola, zda druhá strana ukončila hovor 
    if (response.indexOf("NO CARRIER") != -1) { 
      Serial.println("❌ Hovor byl ukončen druhou stranou."); 
      endCall();  // Ukončení hovoru, pokud druhá strana zavěsí 
    } 
  } 



} 
 
void waitForButtonPress() { 
  while (digitalRead(BUTTON_PIN) == HIGH) { 
    if (incomingCall) blinkLEDNonBlocking(); 
  } 
  delay(200); 
  while (digitalRead(BUTTON_PIN) == LOW); 
  delay(200); 
} 
 
void acceptCall() { 
  Serial.println("✅ Přijímám hovor..."); 
  SIM900.println("ATA"); 
  callActive = true; 
  incomingCall = false; 
  digitalWrite(LED_PIN, LOW); 
} 
 
void endCall() { 
  if (callActive) { 
    Serial.println("❌ Ukončuji hovor..."); 
    SIM900.println("ATH"); // Ukončení hovoru 
    delay(1000); // Krátká pauza pro zpracování příkazu 
 
    digitalWrite(LED_PIN, LOW); 
    callActive = false; 
 
    Serial.println("🧹 Čistím buffer..."); 
    while (SIM900.available()) SIM900.read(); // Vyprázdnění bufferu 
 
    Serial.println("⏳ Čekám na stabilizaci sítě (10s)..."); 
    delay(10000); // Delší zpoždění pro zabránění falešnému RING signálu 
  } 
} 
 
void startCall() { 
  Serial.println("📲 Volám na " + phoneNumber + "..."); 
  SIM900.println("ATD" + phoneNumber + ";"); 
  delay(100); 
  SIM900.println(); 
  callActive = true; 
 
  phoneNumber = "+420"; // Reset čísla po vytočení 
  digitCount = 0; 
} 
 
void blinkLEDNonBlocking() { 
  unsigned long currentMillis = millis(); 



  if (currentMillis - previousMillis >= blinkInterval) { 
    previousMillis = currentMillis; 
    ledState = !ledState; 
    digitalWrite(LED_PIN, ledState); 
  } 
} 

2.6.2 Sériový monitor 

Po zapnutí zařízení se na sériovém monitoru zobrazí zpráva o spuštění. Když sepne kontakt relé 

pro zadávání čísla, systém oznámí, že je možné začít vytáčet číslo. Každá číslice zadaná 

rotačním číselníkem se vypíše spolu s aktuálním stavem čísla. Po zadání 6 číslic s automatickou 

předvolbou +420 se číslo automaticky vytočí. Pokud přijde hovor, zobrazí se „RING“ a čeká 

se na stisk tlačítka pro přijetí. Po jeho stisknutí se hovor přijme a následným stiskem se ukončí. 

Pokud se uvolní tlačítko pro zadávání čísla, číslo se vyresetuje. 

 

Obrázek 9 – informace vypisované na sériový monitor 



 2.7 Testování a kalibrace 
Během fáze testování a kalibrace jsem ověřoval správnou funkčnost všech klíčových částí 

modernizovaného telefonu. Nejprve jsem testoval rotační číselník, zda Arduino správně snímá 

a vyhodnocuje počet impulsů pro jednotlivá čísla. Následovalo ladění přenosu čísel do GSM 

modulu ve formě DTMF signálů, aby odpovídaly skutečnému vytáčenému číslu. Dále jsem 

testoval zvukovou část – tedy přenos hlasu mezi elektretovým mikrofonem a reproduktorem 

přes GSM modul. Pro dosažení srozumitelného zvuku jsem musel kalibrovat hlasitost a citlivost 

mikrofonu, protože původní mikrofonní vstup byl příliš slabý a neodpovídal potřebám modulu. 

Podobně jsem otestoval i zvonek ovládaný servomotorem, kde jsem upravoval úhly otáčení a 

rychlost, aby kladívko správně naráželo do zvonků a zvuk zněl co nejvíce jako původní 

vyzvánění. Celý systém jsem postupně ladil v několika testovacích cyklech, během kterých 

jsem dolaďoval jak hardwarové spojení, tak i programové řízení přes Arduino. Výsledkem byla 

plně funkční reinkarnace historického telefonu schopného uskutečňovat i přijímat hovory v 

moderní mobilní síti. 

 

Obrázek 10 – testování 



3. Závěr 
Cílem této ročníkové práce bylo modernizovat starý telefon Tesla AS20 a přizpůsobit ho pro 

fungování v současné mobilní síti pomocí Arduina Nano a GSM modulu SIM800L. Projekt 

spojil prvky elektroniky, programování, mechanické konstrukce i 3D tisku a výsledkem je plně 

funkční zařízení, které zachovává původní vzhled i ovládání historického telefonu, ale je 

schopné přijímat i uskutečňovat hovory přes moderní technologie. 

Podařilo se mi zachovat rotační číselník, který jsem upravil pro digitální zpracování impulsů, i 

původní zvonek, který nyní ovládá servomotor s pružinkovým kladívkem. Původní uhlíkový 

mikrofon jsem nahradil elektretovým, který je vhodnější pro spolupráci s Arduinem, a celý 

systém jsem propojil prostřednictvím vlastnoručně navrženého zapojení a 3D tištěných dílů. 

Testování a ladění jednotlivých částí potvrdilo funkčnost a stabilitu celého řešení. 

Tato práce mi umožnila propojit historickou techniku s moderním světem a zároveň rozšířit své 

znalosti v oblasti elektroniky, konstrukce, programování a praktického bastlení. Výsledný 

projekt může sloužit jako inspirace pro další způsoby, jak smysluplně oživit stará zařízení a dát 

jim nové využití. 

Do budoucna je možné projekt dále vylepšovat a rozšiřovat – například přidáním malého 

displeje, který by zobrazoval informace z Arduino sériového monitoru (např. vytáčené číslo 

nebo stav modulu), nebo implementací bateriového napájení, díky kterému by telefon mohl být 

přenosný a nezávislý na síťovém napájení. Taková rozšíření by ještě více zvýšila praktičnost a 

technickou úroveň celého zařízení. 

Obrázek 11 – složený telefon 

